
Abstract

This document introduces the syntax and semantics of the language “datalog with stratified nega-
tion,” and shows that programs in this language can be executed in polynomial time. After studying
this document, students should be able to write their own programs in this language.

1 Schema and Database Instance

Datalog is a query language for relational databases. A relational database schema (or schema for short)
is a finite set of relation names with associated arities. We write R/n to indicate that R is a relation
name of arity n.

Example 1.1. In our running example, we will use a schema that contains Knows/2 and Owns/2.

A fact is an expression R(c1, . . . , cn) where each ci is a constant. A database instance I is a finite set of
facts.

Example 1.2. The set I = {Knows(Jeb, Don), Knows(Don, Jeb), Knows(An, Don), Knows(Ed, An),
Owns(Don, iPad), Owns(Don, iPod), Owns(Jeb, iPod)} is a database instance.

The term EDB predicate is often used as a synonym for “relation name of the relational database schema.”
The letter E in EDB stands of Extensional. I like to think about it as Existential : that what really Exists
in the database.

Example 1.3. Knows and Owns are EDB predicates.

As a minor comment, in mathematical logic, one often uses the term predicate symbol for what I call a
predicate in this document.

2 Datalog Syntax

We assume a set of IDB predicates with associated arities. The IDB predicates are distinct from the
EDB predicates. The letter I in IDB stands of Intensional. I like to think about it as Inferred, because
datalog is all about Inferring IDB facts from a database instance of EDB facts.

A fact P (c1, . . . , cn) is an EDB fact if P is an EDB predicate, and is an IDB fact if P is an IDB predicate.
An atom is an expression P (t1, . . . , tn) where each ti is a constant or a variable. The letter t refers to
term, which is a variable or a constant.

A datalog rule is an expression

H(~t0)︸ ︷︷ ︸
rule head

← R1(~t1), . . . , R`(~t`),¬R`+1(~t`+1), . . . ,¬Rm(~tm)︸ ︷︷ ︸
rule body

(1)

where

� H is an IDB predicate,

� each Ri is an EDB predicate or an IDB predicate, and

� each ~ti is a sequence of terms (of appropriate length).

Note that the same predicate can occur more than once in a rule. Within a rule, we distinguish between
the head and the body, as indicated in (1). Every rule must be safe:

1



Safety Requirement: Every variable that occurs in a rule (in its head or its body), must
occur in a nonnegated atom of the rule body.

Informally, the safety requirement guarantees that every variable can be restricted to a finite domain.

Significantly, EDB predicates cannot be used in rule heads. A datalog program P is a finite set of rules,
which can be written in any order.

Example 2.1. Assume IDB predicates Happy/1 and Unhappy/1.

Happy(x) ← Owns(x, iPad),Owns(x, iPod)

Happy(x) ← Knows(x, y),Happy(y)

Happy(x) ← Knows(x, y),¬Knows(x, Don)

Unhappy(x) ← Owns(x, y),¬Happy(x)

It can be easily seen that all rules are safe. In the last rule, the variable x is restricted to the constants
that occur at the first position of some Owns-fact.

Informally, the semantics is as follows. An instantiation of a rule is obtained by replacing all variables
with constants. Such an instantiation is called a ground rule. Then, whenever the body of a ground rule
is true, we infer that its rule head is also true. The body of a ground rule is true if every nonnegated fact
is true, and every negated fact is false.

Example 2.2. Consider the program of Example 2.1. An instantiation of the first rule is:

Happy(Don) ← Owns(Don, iPad),Owns(Don, iPod)

Since Owns(Don, iPad) and Owns(Don, iPod) are in the database instance, they are both true. Therefore,
the body is true, and we can infer that Happy(Don) is true. An instantiation of the second rule is:

Happy(Jeb) ← Knows(Jeb, Don),Happy(Don)

Since Knows(Jeb, Don) is in the database instance, and Happy(Don) has been inferred to be true, we can
infer that Happy(Jeb) is true. An instantiation of the third rule is:

Happy(Ed) ← Knows(Ed, An),¬Knows(Ed, Don)

Since Knows(Ed, An) is in the database instance, and Knows(Ed, Don) is false (because it is not in the
database instance), we can infer that Happy(Ed) is true.

3 Stratified Negation

3.1 The Problem of “Circular” Negation

If P is an EDB predicate, then the truth of a fact P (c1, . . . , cn) is naturally defined as follows:

� P (c1, . . . , cn) is true if it is in the database instance; and

� P (c1, . . . , cn) is false (and therefore ¬P (c1, . . . , cn) is true) if P (c1, . . . , cn) is not in the database
instance.

This way of defining truth is also called the Closed World Assumption (CWA).

If P is an IDB predicate, then it is trickier to determine the truth of a fact P (c1, . . . , cn). Consider the
following program:

Man(x) ← Owns(x, y),¬Female(x)

Female(x) ← Owns(x, y),¬Man(x)

We can obtain the following ground rules:

Man(Jeb) ← Owns(Jeb, iPod),¬Female(Jeb)

Female(Jeb) ← Owns(Jeb, iPod),¬Man(Jeb)

2



Note incidentally that every program without variables is necessarily safe. Since Owns(Jeb, iPod) is true
(because it is in the database), we can simplify these ground rules as follows:

Man(Jeb) ← ¬Female(Jeb)

Female(Jeb) ← ¬Man(Jeb)

This program is cumbersome: the first rule allows us to infer Man(Jeb) if we cannot infer Female(Jeb);
and the second rule allows us to infer Female(Jeb) if we cannot infer Man(Jeb). This leads to an
impossible situation: the first rule cannot be executed before the second rule, and the second rule cannot
be executed before the first rule. Significantly, a datalog program does not depend on the order in which
its rules are listed. So the previous program is really the same as:

Female(Jeb) ← ¬Man(Jeb)

Man(Jeb) ← ¬Female(Jeb)

3.2 Stratified Negation

A stratification of a datalog program assigns a nonnegative integer, called stratum, to every IDB predicate
of the program such that for every program rule

H(~t0) ← R1(~t1), . . . , R`(~t`),¬R`+1(~t`+1), . . . ,¬Rm(~tm)

the following holds:

Stratification Requirement:

� for every i ∈ {1, . . . , `}, if Ri is an IDB predicate, then the stratum of Ri is smaller than
or equal to the stratum of H; and

� for every i ∈ {` + 1, . . . ,m}, if Ri is an IDB predicate, then the stratum of Ri is strictly
smaller than the stratum of H.

That is, if the stratum of the head predicate H is the integer h, then all IDB predicates in the rule
body must have a stratum that is ≤ h. Moreover, negated IDB predicates in the rule body must have a
stratum < h.

Not all datalog programs have a stratification. In particular, we claim that there is no stratification
for:

Man(x) ← Owns(x, y),¬Female(x)

Female(x) ← Owns(x, y),¬Man(x)

To prove our claim, assume that we assign stratum m to Man, and stratum f to Female. Then, the
Stratification Requirement applied to the first rule implies f < m. Likewise, the Stratification Require-
ment applied to the second rule implies m < f . But no stratification can possibly satisfy both f < m
and m < f .

We say that a datalog program is stratified if it has a stratification. From now on, we will reject all
programs that are not stratified. If we have a stratification, we will assume without loss of generality
that the strata are numbered 0, 1, 2 . . .

Example 3.1. We compute a stratification for the program:

Happy(x) ← Owns(x, iPad),Owns(x, iPod)

Happy(x) ← Knows(x, y),Happy(y)

Happy(x) ← Knows(x, y),¬Knows(x, Don)

Unhappy(x) ← Owns(x, y),¬Happy(x)

The Stratification Requirement only talks about IDB predicates, and therefore we can ignore the EDB
predicates Owns and Knows. Let the stratum of Happy be h, and the stratum of Unhappy be u. Because
of the second program rule, we must have h ≤ h, which is vacuously satisfied. Because of the last program
rule, we must have h < u. We obtain a valid stratification by letting h = 0 and u = 1. This stratification
is summarized in the following table:

stratum 0: Happy
stratum 1: Unhappy

3.3 Semantics of Stratified Negation

Assume we have computed a stratification for a stratified datalog program P, and let m be the greatest
stratum in our stratification. For every h ∈ {0, 1, . . . ,m}, we write P(h) for the set of rules whose head
predicate has stratum h.

Example 3.2. For the program P of Example 3.1, the subprograms P(0) and P(1) are as follows:

P(0) =

 Happy(x) ← Owns(x, iPad),Owns(x, iPod)
Happy(x) ← Knows(x, y),Happy(y)
Happy(x) ← Knows(x, y),¬Knows(x, Don)

P(1) =
{

Unhappy(x) ← Owns(x, y),¬Happy(x)

The datalog program is then executed in the following order:

� Execute P(0).

� Execute P(1).

� Execute P(2).

� . . .

� Execute P(m).

When we say “Execute P(h),” we mean to repeatedly instantiate all rules of P(h) in all possible ways.
That is, repeatedly compute ground rules by replacing variables with constants. Whenever the body of
such a ground rule is true, infer that its rule head is also true. When the body of a rule in P(h) contains
a negated atom ¬Ri(~ti), then the Stratification Requirement tells us that Ri is either an IDB predicate
with stratum < h, or an EDB predicate. Therefore, when we execute P(h), all rules that can compute
Ri-facts have already been executed, and we can safely say that a fact ¬Ri(~c) is true if Ri(~c) has not
been inferred. In other words, when executing P(h), we can treat negated predicates as if they were EDB
predicates. We will have more to say about program execution in Section 6.

Note that the rules of a stratified datalog program can be listed in any order. The stratification is the
responsibility of the datalog engine that executes the program.

Example 3.3. The following program computes the complement of the transitive closure of Knows.

Person(x) ← Knows(x, y)

Person(y) ← Knows(x, y)

TransitivelyKnows(x, y) ← Knows(x, y)

TransitivelyKnows(x, y) ← Knows(x, z),TransitivelyKnows(z, y)

NotTransitivelyKnows(x, y) ← Person(x),Person(y),¬TransitivelyKnows(x, y)

Let the strata of Person, TransitivelyKnows, and NotTransitivelyKnows be p, t, and n respectively. The
Stratification Requirement requires p ≤ n and t < n. These inequalities are satisfied by the following
stratification:

stratum 0: Person, TransitivelyKnows
stratum 1: NotTransitivelyKnows

This stratification gives the following programs P(0) and P(1):

P(0) =


Person(x) ← Knows(x, y)
Person(y) ← Knows(x, y)

TransitivelyKnows(x, y) ← Knows(x, y)
TransitivelyKnows(x, y) ← Knows(x, z),TransitivelyKnows(z, y)

P(1) =
{

NotTransitivelyKnows(x, y) ← Person(x),Person(y),¬TransitivelyKnows(x, y)

Therefore, a datalog engine will first execute the rules of P(0) until no more facts can be inferred. Then,
and only then, P(1) is executed. Therefore, when P(1) is executed, all TransitivelyKnows-facts have
already been computed, and therefore we know which TransitivelyKnows-facts are not true.

Note incidentally that a program can have more than one valid stratification. Here is another stratification
for this program:

stratum 0: TransitivelyKnows
stratum 1: Person, NotTransitivelyKnows

It can be shown that the output of a stratified datalog program does not depend on the stratification
that is chosen to execute the program: all stratifications result in the same output.

To conclude, every stratified datalog program can be regarded as a sequence of smaller datalog programs,
one for each stratum, that only use “classical” negation.

Example 3.4. A bus company called Red Cy uses a binary EDB predicate Red/2 to store its direct
bus connections. Moreover, a predicate RedCanceled/2 stores the connections that are currently can-
celed due to a virus outbreak. For example, I = {Red(Mons, Ath), Red(Ath, Dour), Red(Dour, Mons),
Red(Mons, Huy), Red(Ans, Mons), Red(Huy, Ans), Red(Ans, Spa), Red(Spa, Huy),RedCanceled(Ans, Mons)}.
We want to define an IDB predicate CanAlwaysReturn/1 such that CanAlwaysReturn(x) holds true if
on any journey that starts from x, one can always return to x. For our example database instance,
CanAlwaysReturn(Mons) is false, because when one travels from Mons to Huy, one cannot return to Mons.
On the other hand, CanAlwaysReturn(Huy) is true: it is possible to return to Huy from every station
currently reachable from Huy.

When computing a set in datalog, it is sometimes easier to first compute the complement of that set. Let
us introduce CannotAlwaysReturn/1 with the meaning that CannotAlwaysReturn(x) is false if and only
if CanAlwaysReturn(x) is true. So our program will contain the following rules:

CanAlwaysReturn(x) ← Station(x),¬CannotAlwaysReturn(x)

Station(x) ← Red(x, y)

Station(y) ← Red(x, y)

The IDB predicate Station is useful for making the first rule safe. We are now ready to give the rules for
CannotAlwaysReturn(x), which is true if one can travel from x to some station y, but there is no journey
from y back to x.

RedTrip(x, y) ← Red(x, y),¬RedCanceled(x, y)

RedTrip(x, y) ← Red(x, z),RedTrip(z, y),¬RedCanceled(x, z)

CannotAlwaysReturn(x) ← RedTrip(x, y),¬RedTrip(y, x)

A stratification of this program is as follows:

stratum 0: Station, RedTrip
stratum 1: CannotAlwaysReturn
stratum 2: CanAlwaysReturn

This stratification gives the following programs P(0), P(1), and P(2):

P(0) =


Station(x) ← Red(x, y)
Station(y) ← Red(x, y)

RedTrip(x, y) ← Red(x, y),¬RedCanceled(x, y)
RedTrip(x, y) ← Red(x, z),RedTrip(z, y),¬RedCanceled(x, z)

P(1) =
{

CannotAlwaysReturn(x) ← RedTrip(x, y),¬RedTrip(y, x)

P(2) =
{

CanAlwaysReturn(x) ← Station(x),¬CannotAlwaysReturn(x)

Nonetheless, programmers can write the rules of their program in any order, and the output is independent
of the chosen order.

4 Program Dependency Graph

The program dependency graph (PDG) of a datalog program is a directed edge-labeled graph whose
vertices are the IDB predicates of the program. There is a directed edge from H to P if there is a rule
such that H is the head predicate, and P is an IDB predicate that occurs in the body; such an edge is
labeled − if P occurs within the scope of ¬.

For example, for the program of Example 3.4, the PDG is as follows:

RedTrip

CanAlwaysReturn

CannotAlwaysReturn

Station

−

−

It is easy to show the following property:

A program P is stratified if and only if its PDG has no directed cycle that contains an edge
with label −.

Consequently, in the PDG of a stratified datalog program, any directed (possibly infinite) path that starts
from some vertex can traverse at most finitely many edges with label −. It is then easy to show the
following:

Let f be the function that assigns to each IDB predicate P the greatest number of edges with
label − on any directed path that starts from P . We have just argued that this number is
finite. Then, f is a valid stratification. For example, in the preceding PDG, f assigns 2 to
CanAlwaysReturn, 1 to CannotAlwaysReturn, and 0 to both RedTrip and Station.

5 Datalog Languages

In the literature, the name “datalog” is usually used only for programs that do not contain negation.
For programs with stratified negation, one commonly uses “stratified datalog” or “datalog with stratified
negation.”

“Datalog with 6=” refers to datalog without negation, but with 6=. In “datalog with stratified negation,”
we can compute a NotEqual/2 predicate, and therefore “datalog with stratified negation and 6=” is not
more powerful than “datalog with stratified negation.” For example,

Equal(x, x) ← Person(x)

NotEqual(x, y) ← Person(x),Person(y),¬Equal(x, y)

A (stratified) datalog program is called linear if in the body of each rule, there is at most one occurrence
of an IDB predicate that has the same stratum as the rule head.1 For example, the following rule cannot

1Note that, by the Stratification Requirement, such an IDB predicate cannot occur within the scope of ¬ in the rule
body.

occur in a linear datalog program:

TransitivelyKnows(x, y) ← TransitivelyKnows(x, z),TransitivelyKnows(z, y). (2)

A linear datalog program can contain the following rule:

TransitivelyKnows(x, y) ← Knows(x, z),TransitivelyKnows(z, y). (3)

It is generally a good idea to solve a problem by a linear datalog program whenever this is possible,
because such programs can be executed more efficiently than nonlinear programs. Informally, (2) is less
efficient than (3) because it uses two recursive calls.

A program is in semi-positive datalog if negation only applies to EDB predicates.

6 Analysis of Program Execution

In Section 3.3, we pointed out that a program P is executed by successively executing P(0), P(1), P(2),. . . .
In this section, we discuss in more detail the execution of these subprograms. In particular, for any
fixed stratified datalog program P, we will discuss the complexity of the following computational prob-
lem:

INPUT: A database instance.

OUTPUT: All IDB facts that can be inferred by P.

We will show that this problem is in polynomial time in the size of the input. Note that P itself is not
part of the input.

6.1 Guiding Example

We start by an example explaining the execution of P(0) in Example 3.4. This execution is captured by
the following pseudo-code:

Station0 := ∅
RedTrip0 := ∅

i := 0
loop

Stationi+1 := {x | ∃yRed(x, y)}
Stationi+1 := {y | ∃xRed(x, y)}

RedTripi+1 := {x, y | Red(x, y) ∧ ¬RedCanceled(x, y)}
RedTripi+1 := {x, y | ∃z

(
Red(x, z) ∧ RedTripi(z, y) ∧ ¬RedCanceled(x, z)

)
}

i := i + 1
until a fixed point is reached

This pseudo-code initializes the IDB predicates as empty. The loop then repeatedly assigns new values
to the IDB predicates. The queries at the right-hand of := are relational calculus queries, which can
be implemented in relational algebra or SQL, and executed in polynomial time (and even in logarithmic
space). For example, the query

{x, y | ∃z
(
Red(x, z) ∧ RedTripi(z, y) ∧ ¬RedCanceled(x, z)

)
} (4)

is equivalent to (using numbers for attributes):

SELECT Red .1, RedTripi.2

FROM Red , RedTripi

WHERE Red .2 = RedTripi.1
AND NOT EXISTS ( SELECT ∗

FROM RedCanceled
WHERE RedCanceled .1 = Red .1
AND RedCanceled .2 = Red .2 )

It is easy to see:

� RedTrip0 ⊆ RedTrip1; and

Red 1 2
Mons Ath

Ath Dour

Dour Mons

Mons Huy

Ans Mons

Huy Ans

Ans Spa

Spa Huy

RedCanceled 1 2
Ans Mons

Station0 1 RedTrip0 1 2

Station1 1
Mons

Ath

Dour

Huy

Ans

Spa

RedTrip1 1 2
Mons Ath

Ath Dour

Dour Mons

Mons Huy

Huy Ans

Ans Spa

Spa Huy

Station2 1
Mons

Ath

Dour

Huy

Ans

Spa

RedTrip2 1 2
Mons Ath

Ath Dour

Dour Mons

Mons Huy

Huy Ans

Ans Spa

Spa Huy

Mons Dour

Ath Mons

Dour Ath

Dour Huy

Mons Ans

Huy Spa

Ans Huy

Spa Ans
...

...

Figure 1: First two iterations in a fixed point computation.

� if RedTripi ⊆ RedTripi+1, then RedTripi+1 ⊆ RedTripi+2.

Informally, the second item is true because in (4), RedTripi does not occur within the scope of a negation.
From these two items, it follows:

RedTrip0 ⊆ RedTrip1 ⊆ RedTrip2 ⊆ RedTrip3 ⊆ · · ·

The first steps of this computation are shown in Figure 1. Since there are finitely many stations, this
sequence must necessarily reach a fixed point. Note that the sequence for Station already reaches a fixed
point after one iteration.

We now argue that a fixed point will be reached after a number of steps that is polynomial in the size
of the database instance. To this end, assume that the database instance has size `. Then, there can be
at most ` distinct stations, and at most `2 distinct RedTrip-facts. Therefore, it must be the case that

RedTrip`2 = RedTrip`2+1.

6.2 General Treatment

We briefly sketch how the treatment of Section 6.1 generalizes to some P(h), where h is any stratum. A
pseudo-code can be constructed as in Section 6.1, as follows. Assume that P(h) contains a rule

H(~t0) ← R1(~t1), . . . , R`(~t`),¬R`+1(~t`+1), . . . ,¬Rm(~tm),

where the IDB predicates of stratum h are H, R1, R2, . . . , Rk (1 ≤ k ≤ `), while Rk+1, . . . , Rm are either
IDB predicates of a smaller stratum or EDB predicates. In particular, R`+1, . . . , Rm cannot be IDB
predicates of stratum h because of the Stratification Requirement. Then, the loop of the pseudo-code
contains an instruction:

Hi+1 :=

{
~t0

∣∣∣ ∃~x( Ri
1(~t1) ∧ · · · ∧Ri

k(~tk)∧
Rk+1(~tk+1) ∧ · · · ∧R`(~t`) ∧ ¬R`+1(~t`+1) ∧ · · · ∧ ¬Rm(~tm)

)}
, (5)

where ~x contains the variables that are not in ~t0.

Since the IDB predicates of stratum h (i.e., Ri
1, . . . , R

i
k in the query (5)) do not occur within the scope

of a negation, the pseudo-code must necessarily reach a fixed point. After how many steps will this fixed
point be reached? Let n be the arity of H, and assume a database instance of size `. Since such a
database instance can contain at most ` constants, we can generate at most `n distinct H-facts from this
database, which is a polynomial number (because n is fixed). It follows that a fixed point will be reached
in polynomial time in the size of the database. To conclude, the program P(h) runs in polynomial time.
Since the number of strata is constant (it is not part of the input), every stratified datalog program can
be evaluated in polynomial time.

7 Datalog Engine

A datalog engine called DLV can be downloaded at http://www.dlvsystem.com/. To run the program
of Example 3.4, we create the text file brol.txt as shown in Figure 2. Constants start with a lowercase
letter, and variables with an uppercase letter. More information on the DLV syntax can be found at
http://www.dlvsystem.com/html/DLV_User_Manual.html#AEN928. Note that DLV supports a larger
language than stratified datalog with 6=. For example, DLV accepts programs in which negation is not
stratified. However, in the course Bases de Données II, the aim is to use only stratified datalog. When
we execute our program by means of the command

dlv -filter=CanAlwaysReturn brol.txt

we get the following output:

{CanAlwaysReturn(huy), CanAlwaysReturn(ans), CanAlwaysReturn(spa)}

For help, use the command dlv -help.

A common situation is where we want to separate the database from the datalog program, so that the
same database can be used with different programs. The following command executes the program
pgm.txt on the database in db.txt:

dlv db.txt pgm.txt

% This is file brol.txt

% The database.

Red(mons, ath).

Red(ath, dour).

Red(dour, mons).

Red(mons, huy).

Red(ans, mons).

Red(huy, ans).

Red(ans, spa).

Red(spa, huy).

RedCanceled(ans, mons).

% The program.

CanAlwaysReturn(X) :- Station(X), not CannotAlwaysReturn(X).

Station(X) :- Red(X,Y).

Station(Y) :- Red(X,Y).

Redtrip(X,Y) :- Red(X,Y), not RedCanceled(X,Y).

Redtrip(X,Y) :- Red(X,Z), Redtrip(Z,Y), not RedCanceled(X,Z).

CannotAlwaysReturn(X) :- Redtrip(X,Y), not Redtrip(Y,X).

Figure 2: The program of Example 3.4 in DLV syntax.

10


	Schema and Database Instance
	Datalog Syntax
	Stratified Negation
	The Problem of ``Circular'' Negation
	Stratified Negation
	Semantics of Stratified Negation

	Program Dependency Graph
	Datalog Languages
	Analysis of Program Execution
	Guiding Example
	General Treatment

	Datalog Engine

